APPLE CLOCK

Mountain Hardware, Inc.

APPLE CLOCK

Mountain Hardware, Inc.
5523 SCOTTS VALLEY DR., SCOTTS VALLEY, CA 95066

TABLE OF CONTENTS

INDEX OF TABLES

INTRODUCTION

WELCOME TO THE WORLD OF REAL-TIME!

Your Mountain Hardware Apple Clock™ extends the reach of your Apple II* computer by adding the dimension of real time and date in intervals of from 1 millisecond to a little over one year.

On-board battery power keeps the clock running for periods of up to 4 days when your computer is turned off, either intentionally or due to a power outage. If you have down times longer than 4 days, refer to the section in this manual entitled "The Battery".

The Apple Clock is very easy to use because it contains an on-board ROM. This ROM contains software making it easy to obtain the time and date whether you are using Integer BASIC, Applesoft, or Assembly language.

Let your imagination guide you to the many ways in which your Apple Clock can be used.

*Apple II is a trademark of Apple Computer Company.

INSTALLATION

PLUG IN AND GO!

Installing the Apple Clock in your computer is very easy.

1. Turn your computer OFF.

2. Remove the top cover from the computer.

3. Take your Apple Clock and clip the battery connector onto the top of the battery.

4. Plug the Apple Clock into any empty slot on the back of the Apple computer board. You may use any slot EXCEPT SLOT #0. We recommend SLOT #4.

5. Be sure the Apple Clock is firmly seated in its socket.

6. Leave the cover off for the moment. You will need to have access to the switches on the Apple Clock when you set it.

7. Refer to the section in this manual called "Setting the Time".

8. Once the clock has been set, be sure you have changed the 'WRITE PROTECT' switch on the clock according to the instructions in "Setting the Time".

9. YOU ARE FINISHED! You may now replace the cover on your computer.

You are now ready to start reading the time. ONE THING TO REMEMBER - The on-board battery on the clock will take four (4) days to completely charge up. Therefore, leave your computer on continuously for at least four (4) days to completely charge your battery. If this is not done, you cannot be sure that the battery will keep the clock running when you shut your computer off. Once it is charged, however, the clock will keep running for up to 4 days when you shut your computer power off.

THE BATTERY

The Apple Clock is supplied with a rechargeable NiCad battery to keep the clock running when the computer is turned off, or when the power fails. This battery is attached to the Apple Clock on the back side of the board. The battery powers the clock circuitry on the board permitting the clock to keep correct time for periods up to 4 days, if it is fully charged.

To fully charge the battery,
the Apple computer must be left
on for at least 4 days.

This initial charging time of 4 days is designed to maximize the life of the battery. As a general guideline, the battery should charge 2 hours for every 1 hour of use. The battery life is several years, but should be replaced if its performance drops significantly. You may obtain replacements anywhere batteries are sold.

If you anticipate that your computer is going to be turned off for periods longer than 4 days, you may incorporate a larger capacity battery. This can be done by clipping an additional battery clip to the one mounted on the Apple Clock. The two wires from this clip can be run outside the case of the Apple computer and connected to a larger battery. The size of the battery is unimportant, however you must use a battery with a voltage between 7-10V DC.

The battery is intended to support the clock if power fails, or if the computer is turned off for short periods of time. Your computer's lifetime will not be affected by leaving it on continually, and may even be increased. The power consumed by the Apple computer is less than an ordinary light bulb. Consequently we recommend that the Apple computer be left on continually. The clock's battery will keep the clock running if the power fails in your building, or if you turn the computer off for short periods of time (less than 4 days).

SETTING THE TIME

To set the time with the supplied cassette:

1. Load Applesoft into your system.

2. Load the "Set the Time" program supplied with your Apple Clock from cassette. (See Note.)

3. Change the 'WRITE PROTECT' switch to the 'WRITE' position. It is the top switch on the board. Press the switch down on the right side.

4. Set the Leap Year switch. If the current year is a leap year, press the switch on the LEFT. Press the switch down on the RIGHT if it is not a leap year. The leap year switch is the second switch down from the top.

5. Type 'RUN'.

6. Answer the questions that appear on the screen appropriately.

7. After the clock is set, 'WRITE PROTECT' it by pressing the 'WRITE PROTECT' switch down on the LEFT. This prevents the clock from being changed accidently.

The above procedure will need to be performed each January before the 20th of the month. If daylight-saving time is in effect in your area, update the clock as needed.

Store the "Set the Time" cassette in a secure spot for future use. A listing of the "Set the Time" cassette is provided here for your reference. Also listed is the assembly language program used by "Set the Time" (Lines 5ØØØ through 5Ø9Ø).

Note: If you are not using an Applesoft card but are using the Apple Disk, you must type 'CALL 3314' before running the program.

WRITE PROTECT SWITCH

The Apple Clock WRITE PROTECT switch must be in the PROTECT (WP) mode at all times EXCEPT when setting the clock. Otherwise the clock can change time when the computer power is turned off.

Put the switches in these positions when:



SETTING TIME AND IT IS NOT LEAP YEAR



SETTING TIME AND IT IS LEAP YEAR



THE CLOCK HAS BEEN SET AND IT IS NOT LEAP YEAR



THE CLOCK HAS BEEN SET AND IT IS LEAP YEAR

```
1  REM **** MOUNTAIN HARDWARE'S APPLE CLOCK
2  REM **** COPYRIGHT 1978
3  REM ***  SET THE TIME APPLESOFT  ***
4  REM
5  REM **** ADD OR CHANGE THESE LINES FOR DISK SYSTEM
6  REM
7  REM       20 D$=""  WHERE D$="CONTROL D"
8  REM       21 PRINT D$;"NOMON I,O,C"
9  REM     3020 PRINT D$;"PR#";SLOT
10  REM    3025 PRINT D$;"IN#";SLOT
18  REM    3035 PRINT D$;"PR#0"
20  REM    3040 PRINT D$;"IN#0"
24  REM
26  REM
28  REM
30  CALL  - 936
40  VTAB 10
50  PRINT "MOUNTAIN HARDWARE'S APPLE CLOCK"
60  VTAB 13
70  PRINT "DISPLAY OR SET THE TIME PROGRAM"
71 : PRINT : PRINT : PRINT "SEE LINES 5 THRU 10 FOR DI
     SK SYSTEM"
75  PRINT : PRINT
76  INPUT "INPUT THE CLOCK'S SLOT # ";SLOT
80  PRINT : PRINT
90  INPUT "DO YOU WANT TO SET THE TIME (Y ON N)";I$
95  IF I$ = "N" THEN  CALL  - 936: GOTO 2032
100  REM
110  REM  **** POKE IN THE ADVANCEROUTINE AT LOCATION
     $1000
120  REM
130  FOR I = 1 TO 68
140  READ J
150  POKE 767 + I,J
160  NEXT I
300  PRINT
301  PRINT "GIVE THE CURRENT TIME PLUS 30 SECONDS"
302  PRINT
310  INPUT "INPUT THE MONTH (1-12) ";MTH
320  INPUT "INPUT THE DAY (1-31) ";D
330  INPUT "INPUT THE HOUR (0-23) ";H
340  INPUT "INPUT THE MINUTE (0-59) ";M
350  INPUT "INPUT THE SECONDS (0-59) ";S
360  PRINT
365  PRINT "HIT RETURN WHEN YOU HAVE SET THE LEAP"
370  PRINT "SWITCH CORRECTLY, AND ARE SWITCHED FOR"
380  INPUT "WRITING TO THE CLOCK ";I$
390  PRINT : PRINT
```

```
500  REM
501  REM **** CHECK LEAP YEAR SWITCH
502  REM
505  REM   IF L=1 THEN IT'S A LEAP YEAR
510 L =  PEEK (49280 + 16 * SLOT)
511 L =  INT (L / 64)
512  IF L > 1 THEN L = L - 2
600  REM
601  REM **** FIND DAYS TO DATE -- DTD --
602  REM
605 DTD = 0
610  FOR I = 1 TO MTH
620  READ J
630 DTD = DTD + J
640  NEXT I
650 DTD = DTD + D - 1
660  IF L = 1 AND MTH > 2 THEN DTD = DTD + 1
700  REM
701  REM **** CALCULATE SECONDS TO DATE --STD --
702  REM
710 STD = DTD * 86400 + H * 3600 + M * 60 + S
800  REM
801  REM **** PREPARE SECONDS FOR CLOCK
802  REM
810 TEMP = 896: REM  RAM STORAGE AREA
820 S0 =  INT (STD / 2 ^ 20)
825  POKE TEMP,S0
830 STD = STD - S0 * 2 ^ 20
840 S1 =  INT (STD / 2 ^ 12)
850  POKE TEMP + 1,S1
860 STD = STD - S1 * 2 ^ 12
870 S2 =  INT (STD / 2 ^ 4)
880  POKE TEMP + 2,S2
890 STD = STD - S2 * 2 ^ 4
900  POKE TEMP + 3,STD * 16
910  REM
911  REM **** ALSO SAVE N2 AND N7
912  REM
920  POKE TEMP + 4,SLOT * 16 + 2
930  POKE TEMP + 5,SLOT * 16 + 7
1000  REM
1001  REM **** STOP CLOCK AND CALL ADVANCE ROUTINE
1002  REM
1005 SR = 49280 + SLOT * 16 + 5
1006 SP = 49280 + SLOT * 16 + 6
1010 I =  PEEK (SP)
1020  CALL 768: REM  CALL THE MACHINE LANG ADVANCE ROU
     TINE
2000  INPUT "HIT RETURN AT EXACT TIME ";I$
2010 I =  PEEK (SR): REM   START CLOCK
2020  CALL  - 936
2030  PRINT "DON'T FORGET TO WRITE PROTECT THE CLOCK"
2031  VTAB 24
2032  VTAB 24: PRINT "      HIT RESET TO STOP PROGRAM"
```

```
3000  REM
3001  REM **** DISPLAY THE TIME
3002  REM
3005  VTAB 6: PRINT "  MOUNTAIN HARDWARE'S APPLE CLOCK
    "
3006  PRINT : PRINT "            THE TIME IS"
3010  VTAB 22
3020  IN# SLOT
3025  PR# SLOT
3030  INPUT " ";I$
3040  IN# 0
3045  PR# 0
3050  VTAB 12: HTAB 8
3060  PRINT "  ";I$
3070  GOTO 3010
4999  REM **** ADVANCE SUBROUTINE DATA
5000  REM
5010  DATA 72,8,138,72,152,72,174,132
5020  DATA 3,172,133,3,189,128,192,205
5030  DATA 130,3,208,8,189,129,192,205
5040  DATA 131,3,240,6,185,128,192,76
5050  DATA 12,3,202,202,200,189,128,192
5060  DATA 41,31,205,128,3,208,8,189
5070  DATA 129,192,205,129,3,240,6,185
5080  DATA 128,192,76,37,3,104,168,104
5090  DATA 170,40,104,96
5999  REM **** MONTH DATA
6000  DATA  0,31,28,31,30,31,30,31,31,30,31,30
```

```
                        1 * MOUNTAIN HARDWARE'S
                        2 * APPLE CLOCK
                        3 * ADVANCE ROUTINE
                        4 * FOR SETTING THE CLOCK
                        5 * COPYRIGHT 1978
                        6 * GARY MUHONEN
                        7 *
                        8 * SYSTEM EQUATES
                        9 *
                       10 DS      EQU   $C080    ;DEVICE SELECT
                       11 DS1     EQU   $C081    ;DEV SEL +1
                       12 T0      EQU   $0380    ;TEMP STORAGE
                       13 T1      EQU   $0381    ;  FOR TIME
                       14 T2      EQU   $0382    ;  COMPARISON
                       15 T3      EQU   $0383
                       16 T4      EQU   $0384
                       17 T5      EQU   $0385
                       18 *
                       19 * PROGRAM STARTS AT $0300
                       20 *
                       21         ORG   $0300
                       22         OBJ   $5000
                       23 *
                       24 * SAVE REGISTERS
                       25 *
0300 48                26         PHA
0301 08                27         PHP
0302 8A                28         TXA
0303 48                29         PHA
0304 98                30         TYA
0305 48                31         PHA
                       32 *
                       33 *CHECK LOWEST TWO BYTES OF TIME
                       34 *AND INCREMENT AS NECESSARY
                       35 *
0306 AE 84 03          36 I2      LDX   T4       ;N2
0309 AC 85 03          37         LDY   T5       ;N7 FOR ADV1
030C BD 80 C0          38 C2      LDA   DS,X     ;LOAD TIME2 FROM CLOCK
030F CD 82 03          39         CMP   T2       ;DOES IT EQUAL DESIRED?
0312 D0 08             40         BNE   A1       ;NO, GO ADVANCE
0314 BD 81 C0          41         LDA   DS1,X    ;YES, NOW LOAD TIME3
0317 CD 83 03          42         CMP   T3       ;DOES IT EQUAL DESIRED
031A F0 06             43         BEQ   I0       ;YES, GO ON TO NEXT TIME
031C B9 80 C0          44 A1      LDA   DS,Y     ;NO, ADVANCE CLOCK (ADV1)
031F 4C 0C 03          45         JMP   C2       ;GO BACK TO CHECK TILL TIME DESIRED
```

```
                        46 *
                        47 *CHECK HIGHEST TWO BYTES OF TIME
                        48 *AND INCREMENT AS NECESSARY
                        49 *
0322 CA                 50 I0      DEX    ;SET X TO TIME0 SPOT
0323 CA                 51         DEX
0324 C8                 52         INY    ;SET Y TO ADV2
0325 BD 80 C0           53 C0      LDA    DS,X    ;LOAD TIME0
0328 29 1F              54         AND    #$1F    ;STRIP OFF LEAP YEAR AND INTRPT BITS
032A CD 80 03           55         CMP    T0      ;DOES IT EQUAL DESIRED?
032D D0 08              56         BNE    A2      ;NO, GO ADVANCE
032F BD 81 C0           57         LDA    DS1,X   ;YES, CHECK TIME1
0332 CD 81 03           58         CMP    T1      ;DOES IT EQUAL DESIRED?
0335 F0 06              59         BEQ    END     ;YES, WE'RE DONE
0337 B9 80 C0           60 A2      LDA    DS,Y    ;NO, ADVANCE CLOCK (ADV2)
033A 4C 25 03           61         JMP    C0      ;GO BACK AND TRY AGAIN
                        62 *
                        63 *DONE,  SO RECOVER REGS AND RETURN
                        64 *
033D 68                 65 END     PLA
033E A8                 66         TAY
033F 68                 67         PLA
0340 AA                 68         TAX
0341 28                 69         PLP
0342 68                 70         PLA
0343 60                 71         RTS

--- END ASSEMBLY ---
```

READING THE TIME

QUICK READS

The following tricks may be used to quickly read the time. They simply print the time on the screen. TRY THEM!

		CTRL
MONITOR	* n	K
INTEGER	IN#n	
APPLESOFT	IN#n	

Hit RESET to stop.
n = the clock's slot number (1-7)

Table 1
QUICK READS

To get back into BASIC after hitting RESET, use one of the following procedures.

BASIC	COMMAND CTRL
Integer	C
Integer with DOS*	3DØG
Integer with Applesoft Card (Switch down)	CTRL C
Integer with Applesoft Card (Switch down) and DOS	3DØG
Applesoft - Cassette	ØG
Applesoft and DOS	3DØG
Applesoft Card (Switch up) and DOS	3DØG

Table 2
RE-ENTRY PROCEDURE

*DOS stands for Disk Operating System.

DISPLAYING THE TIME

The following four programs display the date and time as one line centered on the screen. Take the time to become familiar with these programs. They can easily be incorporated into programs you write later. We suggest you read the sections on Strings in the Applesoft BASIC Manual and the Apple II BASIC Programming Manual.

It is generally a good practice to set the SLOT number the clock is in at the very beginning of a program. Line 1Ø of the program demonstrates this. Elsewhere in the program, use SLOT instead of the number (2,3...). Later, if you move the clock to a different slot, you need change only one line, instead of searching for all the places where the number was specified.

When using Integer BASIC, it is necessary to dimension the strings. Applesoft does not require this.

Lines 4Ø through 9Ø of the Integer BASIC program without DOS should be used when reading the time from the clock. Leaving out Line 5Ø (PR#SLOT) will cause the time to be printed on the screen when an INPUT (Line 7Ø) is done. The time (T$) is passed to BASIC in the following format:

```
MONTH/DAY HOUR;MINUTE;SECOND.FRACTION
03/04 10;13;14.123
```

When actually doing the input statement (INPUT " ", T$), note that a space is printed to the clock board. This is used so that the data returned from the INPUT statement is the same between Applesoft and Integer BASIC.

There is one difference between the INPUT statement in Applesoft and Integer BASIC.

With Integer BASIC, use

```
INPUT " ", T$
```

With Applesoft, use

```
INPUT " "; T$
```

```
  0 REM  ************  TIME  ************
  3 REM   *** INTEGER BASIC WITHOUT DOS ***
  5 REM
 10 SLOT=4: REM  SET THE SLOT#
 20 DIM T$(25): REM  DIMENSION THE TIME STRING
 30 CALL -936: REM  CLEAR THE SCREEN
 40 IN#SLOT: REM  SET INPUT TO CLOCK BOARD
 50 PR#SLOT: REM  SET OUTPUT TO CLOCK
 60 VTAB 23: REM  PUT CURSOR AT BOTTOM OF SCREEN
 70 INPUT " ",T$: REM   OBTAIN THE TIME
 80 IN#0: REM  RESTORE INPUT TO KEYBOARD
 90 PR#0: REM  RESTORE OUTPUT TO CRT
100 VTAB 12: TAB 10: REM  CENTER THE OUTPUT
110 PRINT T$: REM  OUTPUT TIME TO THE SCREEN
120 GOTO 40: REM  READ TIME AGAIN
```

```
  0  REM **********  TIME  **********
  4  REM  *** APPLESOFT WITHOUT DOS ***
  5  REM
 10 SLOT = 4: REM  SET THE SLOT NUMBER
 20  HOME : REM  CLEAR THE SCREEN
 30  IN# SLOT: REM  SET INPUT TO CLOCK BOARD
 40  PR# SLOT: REM  SET OUTPUT TO CLOCK BOARD
 50  VTAB 23: REM  PUT CURSOR AT BOTTOM OF SCREEN
 60  INPUT " ";T$: REM  OBTAIN THE TIME
 70  IN# 0: REM  RESTORE INPUT TO KEYBOARD
 80  PR# 0: REM  RESTORE OUTPUT TO CRT
 90  VTAB 12: HTAB 10: REM  CENTER THE OUTPUT
100  PRINT T$: REM  OUTPUT THE TIME
110  GOTO 30: REM  READ TIME AGAIN
```

Things change when DOS (Disk Operating System) is active at the time you are running the program. Read your Disk Operating Manual, especially, "Use of the Disk Operating System From Within a Program".

A good programming technique is to set D$ equal to a CONTROL D, as in Line 2Ø. Then use D$ wherever a CONTROL D is required. Line 3Ø prevents the commands IN# and PR# from being printed to the screen when they are executed.

Compare the programs with DOS and without DOS as you will probably be using both versions. The main difference is that when DOS is active, the IN and PR statements are formatted differently.

```
  0 REM ***********  TIME  ***********
  4 REM *** INTEGER BASIC WITH DOS ***
  5 REM
 10 SLOT=4: REM  SET CLOCK SLOT#
 20 D$="": REM  D$="CONTROL D"
 30 PRINT D$;"NOMONI,O,C": REM  PREVENT DISK COMMAND FROM
     PRINTING ON SCREEN
 35 DIM T$(25): REM  DIMENSION TIME ARRAY
 40 CALL -936: REM  CLEAR THE SCREEN
 50 PRINT D$;"IN#";SLOT: REM  SET INPUT TO CLOCK BOARD
 60 PRINT D$;"PR#";SLOT: REM  SET OUTPUT TO CLOCK BOARD
 70 VTAB 23: REM  PUT CURSOR AT BOTTOM OF SCREEN
 80 INPUT " ",T$: REM   OBTAIN THE TIME
 90 PRINT D$;"IN#0": REM  RESTORE INPUT TO KEYBOARD
100 PRINT D$;"PR#0": REM  RESTORE OUTPUT TO CRT
110 VTAB 12: TAB 10: REM  CENTER OUPUT
120 PRINT T$: REM  OUTPUT TIME TO SCREEN
130 GOTO 50: REM  READ TIME AGAIN
```

```
 0  REM  *********  TIME  *********
 4  REM  *** APPLESOFT WITH DOS ***
 5  REM
10 SLOT = 4: REM  SET THE SLOT NUMBER
20 D$ = "": REM  D$=" CONTROL D"
25  PRINT D$;"NOMONI,O,C": REM  KEEP DISK COMMANDS FRO
     M PRINTING
30  HOME : REM  CLEAR THE SCREEN
40  PRINT D$;"IN#";SLOT: REM  SET INPUT TO CLOCK
50  PRINT D$;"PR#";SLOT: REM   SET OUTPUT TO CLOCK
60  VTAB 23: REM  PUT CURSOR AT BOTTOM OF SCREEN
70  INPUT " ";T$: REM  OBTAIN THE TIME
80  PRINT D$;"IN#0": REM  RESTORE INPUT TO KEYBOARD
90  PRINT D$;"PR#0": REM  RESTORE OUTPUT TO KEYBOARD
100  VTAB 12: HTAB 10: REM  CENTER THE OUPUT
110  PRINT T$: REM  OUTPUT TIME TO SCREEN
120  GOTO 40: REM  READ TIME AGAIN
```

OTHER DISPLAY FORMATS

The previous programs simply print the time as it is given by the clock. At times, it will be desireable to use different formats or to only use part of the time. The DATE AND TIME programs here print the time as:

```
DATE: OCTOBER 31, 1978
TIME: 12:3Ø:45.923
```

The time is read (Line 5Ø to 13Ø) just as it was read in the previous section. Now, however, string manipulation is done to the time (T$). In the Integer BASIC program Lines 16Ø through 21Ø show how to find just the month, day, etc. They are repeated here for convenience.

```
MONTH$=T$(1,2)
DAY$=T$(4,5)
HOUR$ T$(7,8)
MINUTES=T$(10,11)
SECONDS=T$(13,14)
FRAC$=T$(16,18)
```

Table 3
INTEGER TIME STRING MANIPULATION

The various components of T$ can then be manipulated to obtain the desired results. Lines 23Ø to 37Ø test the month string to determine the name of the month. The date and time can then be printed out in whatever format is desired.

The Applesoft program is slightly different. The time (T$), however, is read in the same manner (Lines 5Ø through 13Ø). To obtain the elements of T$, the following string manipulations are done.

```
MTH$=LEFT$(T$,2)
DAY$=MID$(T$,4,2)
HOUR$=MID$(T$,7,2)
MINUTE$=MID$(T$,10,2)
SEC$=MID$(T$,13,1)
FRAC$=RIGHT$(T$,3)
```

Table 4
APPLESOFT TIME STRING MANIPULATION

```
  0 REM ********  DATE AND TIME  ********
  4 REM *** INTEGER BASIC WITHOUT DOS ***
  5 REM
 10 DIM T$(25),MONTH$(10),DAY$(2),HOUR$(2),MINUTE$(2
    ),SECOND$(2),FRAC$(3),YEAR$(4): REM  DIMENSION STRING
    S
 20 SLOT=4: REM  SET SLOT NUMBER
 25 YEAR$="1978": REM  SET THE YEAR
 30 CALL -936: REM  CLEAR THE SCREEN
 40 REM
 50 REM  READ THE TIME
 60 REM
 80 IN#SLOT: REM   SET INPUT TO CLOCK BOARD
 90 PR#SLOT: REM   SET OUTPUT TO CLOCK BOARD
100 VTAB 23: REM   PUT CURSOR AT BOTTOM OF SCREEN
110 INPUT " ",T$: REM   OBTAIN THE TIME
120 IN#0: REM   RESTORE INPUT TO KEYBOARD
130 PR#0: REM   RESTORE OUTPUT TO CRT
135 REM
140 REM  OBTAIN MONTH, DAY, HOUR,...ECT
150 REM
160 MONTH$=T$(1,2)
170 DAY$=T$(4,5)
180 HOUR$=T$(7,8)
190 MINUTE$=T$(10,11)
200 SECOND$=T$(13,14)
210 FRAC$=T$(16,18)
220 REM
230 REM  OBTAIN MONTH (JANUARY, FEBRUARY...)
240 REM
250 IF MONTH$="01" THEN MONTH$="JANUARY"
260 IF MONTH$="02" THEN MONTH$="FEBRUARY"
270 IF MONTH$="03" THEN MONTH$="MARCH"
280 IF MONTH$="04" THEN MONTH$="APRIL"
290 IF MONTH$="05" THEN MONTH$="MAY"
300 IF MONTH$="06" THEN MONTH$="JUNE"
310 IF MONTH$="07" THEN MONTH$="JULY"
320 IF MONTH$="08" THEN MONTH$="AUGUST"
330 IF MONTH$="09" THEN MONTH$="SEPTEMBER"
340 IF MONTH$="10" THEN MONTH$="OCTOBER"
350 IF MONTH$="11" THEN MONTH$="NOVEMBER"
360 IF MONTH$="12" THEN MONTH$="DECEMBER"
370 REM
380 REM  PRINT DATE AND TIME ON SCREEN
390 REM
400 VTAB 10: TAB 10: REM    CENTER OUTPUT
410 PRINT "DATE: ";MONTH$;" ";DAY$;", ";YEAR$
420 VTAB 12: TAB 10: REM  CENTER OUTPUT
430 PRINT "TIME: ";HOUR$;":";MINUTE$;":";SECOND$;"."
    ;FRAC$
440 GOTO 80: REM   READ TIME AGAIN
```

```
0  REM ******  DATE AND TIME  ******
4  REM *** APPLESOFT WITHOUT DOS ***
10  REM
20 SLOT = 4: REM  SET SLOT#
30 YEAR$ = "1978": REM  SET YEAR
40  HOME : REM  CLEAR SCREEN
50  REM
60  REM  READ THE TIME
70  REM
80  IN# SLOT: REM  SET INPUT TO CLOCK
90  PR# SLOT: REM  SET OUTPUT TO CLOCK
100  VTAB 23: REM  PUT CURSOR AT BOTTOM OF SCREEN
110  INPUT " ";T$: REM  OBTAIN TIME
120  IN# 0: REM  RESTORE INPUT TO KEYBOARD
130  PR# 0: REM  RESTORE OUTPUT TO CRT
200  REM
210  REM  OBTAIN MONTH, DAY, HOUR,...ECT
220  REM
230 MTH$ =  LEFT$ (T$,2)
240 DAY$ =  MID$ (T$,4,2)
250 HOUR$ =  MID$ (T$,7,2)
260 MINUTE$ =  MID$ (T$,10,2)
270 SEC$ =  MID$ (T$,13,2)
280 FRAC$ =  RIGHT$ (T$,3)
300  REM
310  REM   OBTAIN MONTH (JANUARY, FEBRUARY...)
320  REM
330 MTH =  VAL (MTH$): REM  FIND DECIMAL # FOR MONTH
340  RESTORE : REM  INITIALIZE DATA
350  FOR I = 1 TO MTH
360  READ MTH$: REM  FIND NAME OF MONTH
370  NEXT I
380  DATA  "JANUARY","FEBRUARY","MARCH","APRIL","MAY","JUNE"
390  DATA  "JULY","AUGUST","SEPTEMBER","OCTOBER","NOVEMBER","DECEMBER"
400  REM
410  REM  OUTPUT DATE AND TIME
420  REM
430  VTAB 10: HTAB 10: REM  CENTER OUTPUT
440  PRINT "DATE: ";MTH$;" ";DAY$;", ";YEAR$
450  VTAB 12: HTAB 10: REM  CENTER OUTPUT
460  PRINT "TIME: ";HOUR$;":";MINUTE$;":";SEC$;".";FRAC$
470  GOTO 80: REM  READ TIME AGAIN
```

Obtaining the name of the month in Applesoft is much easier since it is possible to change a string to its decimal value (Line 330). Then data can be read until the correct month is found (Lines 340 through 390). The time and date may then be printed in the desired format.

To print the time as AM or PM, add these lines to the Applesoft program.

```
290 HOUR =  VAL (HOUR$): REM CHANGE HOUR$ TO DECIMAL
291 HR = HOUR
292  IF HR = 0 THEN HOUR = 12
293  IF HR > 12 THEN HOUR = HR - 12
294 AMPM$ = "AM"
295  IF HR > 11 THEN AMPM$ = "PM"
460  PRINT "TIME: ";HOUR;":";MINUTE$;":";SEC$;".";FRAC$
     ;"  ";AMPM$;"    "
```

Table 5

APPLESOFT AM/PM

To print the time as AM or PM, add or change these lines in the Integer BASIC program.

```
 15 DIM HR10$(2),HR1$(2),AMPM$(2)
362 HR10$=HOUR$(1)
363 HR1$=HOUR$(2)
365 HR10= ASC(HR10$)-176
366 HR1= ASC(HR1$)-176
367 HOUR=(10*HR10)+HR1
368 HR=HOUR
369 AMPM$="AM"
370 IF HR=0 THEN HOUR=12
371 IF HR>12 THEN HOUR=HR-12
372 IF HR>11 THEN AMPM$="PM"
430 PRINT "TIME: ";HOUR;":";MINUTE$;":";SECOND$;".";
    FRAC$;"  ";AMPM$;"    "
```

Table 6

INTEGER AM/PM

If the previous programs are going to be run with DOS, change or add these lines.

```
 15 D$="": REM   D$="CONTROL D"
 16 PRINT D$;"NOMONI,O,C": REM   PREVENT DISK
     COMMAND FROM PRINTING ON SCREEN
 80 PRINT D$;"IN#";SLOT: REM   SET INPUT TO C
    LOCK BOARD
 90 PRINT D$;"PR#";SLOT: REM   SET OUTPUT TO
    CLOCK BOARD
110 INPUT " ",T$: REM    OBTAIN THE TIME
120 PRINT D$;"IN#0": REM    RESTORE INPUT TO
    KEYBOARD
130 PRINT D$;"PR#0": REM   RESTORE OUTPUT TO
    CRT
```

Table 7
PROGRAM CHANGES WITH DOS

ELAPSED TIME

Using the Apple Clock, simple programs may be written to measure the elapsed time between two events.

At the initial event, read the clock and save as a string, maybe T1$. At the second event, read the time as T2$. Then, using the subroutine in the following program located at location 3ØØØ, the total seconds to date (STD) since January 1, can be found for each of the two times. If they are subtracted, the elapsed time in seconds is easily found. These seconds can be changed to Days, Hours, Minutes and Seconds by the subroutine at Location 4ØØØ.

The following program measures the time between two carriage returns typed.

```
0  REM **** ELAPSED TIMER PROGRAM ****
2  REM **** APPLESOFT WITH DOS ****
20  REM  *** SUBROUTINES MAY BE USED IN YOUR PROGRAMS
25  REM
30 D$ = "": REM  D$=" CONTROL D"
40  PRINT D$;"NOMON I,O,C": REM  DON'T PRINT DISK COMM
     ANDS
50  HOME
60 SLOT = 4: REM  SET CLOCK BOARD SLOT#
65  REM
70  REM  IF L=0 NOT A LEAP YEAR, IF L=1 IT IS A LEAP Y
     EAR
80 L =  PEEK (49280 + 16 * SLOT)
90 L =  INT (L / 64)
95  IF L > 1 THEN L = L - 2
96  REM
100  REM    T1$=THE INITIAL START TIME
110  REM    T2$=THE TIME AT A LATER TIME
120  INPUT "HIT RETURN TO START TIMER ";A$
125  PRINT
130  GOSUB 2000: REM  GET THE TIME
140 T1$ = T$: REM  T1$=INITIAL START TIME
150  INPUT "HIT RETURN AT DESIRED TEST TIME ";A$
155  PRINT
160  GOSUB 2000: REM  GET THE TIME NOW
170 T2$ = T$: REM  T2$=THE TEST TIME
200  REM
210  REM  FIND STD FOR T1$
220 T$ = T1$: GOSUB 3000:S1 = STD
230  REM  FIND STD FOR T2$
240 T$ = T2$: GOSUB 3000:S2 = STD
250  REM  FIND ELAPSED TIME  ET=S2-S1
260 ET = S2 - S1
290  VTAB 10
300  PRINT "THE ELAPSED TIME HAS BEEN "
310  PRINT ET;" SECONDS"
```

```
500  REM  CONVERT TO DAYS, HOURS MINUTES, SEDONDS
510  GOSUB 4000: REM  SUBR TO CALC THIS
515  VTAB 16
520  PRINT "DAYS=";D
530  PRINT "HOURS=";H
540  PRINT "MINUTES=";M
550  PRINT "SECONDS=";S
600  END
2000  REM
2005  REM  *** SUBR - GET THE TIME
2010  REM *** THESE NEED TO BE CHANGED IF DISK IS NOT
     USED
2030  PRINT D$;"IN#";SLOT
2040  PRINT D$;"PR#";SLOT
2050  INPUT " ";T$
2060  PRINT D$;"IN#0"
2070  PRINT D$;"PR#0"
2080  RETURN
3000  REM
3005  REM  SUBR - STD
3006  REM
3010  REM  CALCULATE SECONDS TO DATE FOR EACH TIME (ST
     D)
3020  REM  THIS IS THE NUMBER OF SECONDS SINCE JANUARY
      1
3030  REM  DO THIS FOR STRING TIME T$
3040  REM    RETURN A NUMBER - STD
3050  REM
3060  REM  FIND #'S FOR DATE AND TIME
3070 MT =  VAL ( MID$ (T$,1,2))
3080 D =  VAL ( MID$ (T$,4,2))
3090 H =  VAL ( MID$ (T$,7,2))
3100 M =  VAL ( MID$ (T$,10,2))
3110 S =  VAL ( MID$ (T$,13,6))
3130  REM  CALCULATE DAYS TO DATE - DTD
3135  RESTORE
3140 DTD = 0
3150  FOR I = 1 TO MT
3160  READ J
3170 DTD = DTD + J
3180  NEXT I
3200  DATA  0,31,28,31,30,31,30,31,31,30,31,30,31
3205  REM   ADD IN DAYS AND LEAP YEAR DAY
3210 DTD = DTD + D
3230  IF MT > 2 AND L = 1 THEN DTD = DTD + 1
3240  REM  FIND SECONDS TO DATE - STD
3250 STD = DTD * 86400 + H * 3600 + M * 60 + S
3300  RETURN
4000  REM
4010  REM  SUBR - PUT SECONDS INTO DAYS, HOURS, MINUTE
     S, SECONDS
4020  REM  GIVEN ET IN SECONDS
4040 D =  INT (ET / 86400)
4050 ET = ET - D * 86400
4060 H =  INT (ET / 3600)
4070 ET = ET - H * 3600
4080 M =  INT (ET / 60)
4090 S = ET - M * 60
4100  RETURN
```

INTERVAL TIMER

In some applications it may be necessary to perform a task at a particular time.

The method to do this is as follows:

1. Obtain the current time from the clock.

2. Convert to seconds to date (STD) using the subroutine in the previous program.

3. Add to it the desired wait time in seconds, and save this time.

4. At various points in your program, check the time and find the current STD.

5. If the current STD is equal or greater than the desired time, it's time to perform the desired task.

THEORY OF OPERATION

THE HARDWARE

The Apple Clock hardware design is composed of four main sections: The clock counters, the PROM circuitry, the supply regulator, and the interrupt hardware.

Clock Counters

The counters are placed across the top of the PC board. A 1 MHz crystal controls the frequency. Three dual BCD up counters (U1, U2, U3) are used to divide the frequency down to obtain the units of time less than a second. Therefore, 1's 1Ø's and 1ØØ's of milliseconds are available in BCD format.

Two 12-bit binary counters (U^4, U5) provide the digits of 2^0 to 2^{23} seconds. A D-flipflop (U10) adds the last time digit of 2^{24} seconds.

A decoder (U13) is used to determine which digit is being read. It controls the enable lines on the Tri-State buffers ($U14^4$-U21).

The clock is set by first stopping the clock (reading from CØ8Ø+N6). Flipflop, U10, keeps the clock stopped until a START clock command is issued. Once the clock is stopped, the ADVANCE 1 command advances the first 12-bit binary counter. An ADVANCE 2 command advances the counters U^4 and U10. Digits below a second cannot be set. They are automatically reset to zero when the clock is stopped. It is impossible to advance the clock when the clock is running. A 'WRITE PROTECT' switch on Pin 9 of U1Ø prevents the clock from being accidentally changed.

ROM Access

An on-board ROM, U27, provides easily accessible software for the user. The flipflop made of U8 and U23 is set by a READ from CNØØ-CNFF. This causes Pin 11 of U8 to go high and stay there until the flipflop (U8, U23) is reset. The ROM can then be read from C8ØØ-CF7F. The ROM is shut off when a $CFFF is addressed (U23), or when 'RESET' is hit.

The output buffers (U25, U26) are enabled by U6 which is dependent upon DEVICE SELECT, I/O SELECT, R/W or PROM ENABLE.

Regulator Circuit

There are two 5 volt supplies for the Apple Clock. One is derived from the 5 volts on the Apple bus. It is used to supply power to all the TTL circuits on the board. The other supply is derived from the +12 volts on the bus and regulated to +5 volts by U28. This regulator supplies the power for all the CMOS circuitry. When the Apple is turned off, the on-board battery supplies the CMOS circuitry, and keeps the clock running.

When the Apple is turned on, the battery is trickle-charged through R12. For faster charging time, R12 may be reduced at the expense of a shorter battery lifetime.

Interrupts

The Apple Clock is capable of generating interrupts on a regular basis. This means that the computer can be performing one task, and be interrupted to perform another task, then return back to the original program.

Interrupts may be enabled by writing a '1' to the set interrupt address, and disabled by writing an 'Ø' to the same address. Hitting 'RESET' will also disable interrupts.

The clock board will interrupt on one-second intervals when enabled. This one-second period is determined by which counter output drives Pin 11 of U11. The clock is factory set for 1-second intervals.

The procedure for handling interrupts is as follows:

1. Enable interrupts by writing a '1' to the SET INTERRUPT device select address.

2. Once a second, an interrupt will occur that will lower the IRQ line on the Apple bus if a higher priority peripheral (in a lower slot number) is not currently interrupting. The 'INT IN' line tells the clock board if a higher priority board is interrupting.

3. Along with the IRQ line going low, the INT OUT line will go low to tell lower priority boards that the clock is interrupting. This prevents them from interrupting.

4. If a higher priority board is interrupting, the clock will wait till it is done, and then the clock will interrupt.

5. If interrupts are enabled in software (CLI instruction), the Apple will perform a jump to an address contained in memory locations 3FE (low) and 3FF (high).

6. An interrupt automatically disables interrupts so that other interrupts may not occur immediately. Two forms of interrupt acknowledges must be performed to the Apple Clock to clear interrupts. One is 'CLEAR IRQ' which clears the IRQ line but not the 'INT-OUT' line. This procedure may be done to allow higher priority peripheral boards to interrupt, but not lower priority boards. To allow higher priority boards to interrupt, do a 'CLEAR IRQ' command early in the interrupt routine and do a 'CLEAR INT-OUT' command at the end of the interrupt routine to allow lower priority boards to interrupt. Before leaving the interrupt routine, both CLEARS should be performed so that other interrupting boards are not tied up and may perform their own interrupts.

7. To prevent the Apple Clock from interrupting, a 'Ø' may be written to the Set Interrupt address, or 'RESET' may be pressed.

THE SOFTWARE

The Apple II peripheral bus is memory mapped. Therefore, in order to talk to a particular device you must address its DEVICE SELECT address. The following table shows the relationship between the slot # the clock is in and the DEVICE SELECT address.

SLOT #	DEVICE SELECT HEX	ADDRESS DECIMAL
Ø	CØ8Ø-CØ8F	(-16256)-(-16241)
1	CØ9Ø-CØ9F	(-16240)-(-16225)
2	CØAØ-CØAF	(-16224)-(-16209)
3	CØBØ-CØBF	(-16208)-(-16193)
4	CØCØ-CØCF	(-16192)-(-16177)
5	CØDØ-CØDF	(-16176)-(-16161)
6	CØEØ-CØEF	(-16160)-(-16145)
7	CØFØ-CØFF	(-16144)-(-16129)

Table 8

DEVICE SELECT ADDRESSING

The following formula may also be used:

DEVICE SELECT ADDRESS = $CØ8Ø+$NX
= -16256+(16*N)+X

N = SLOT #

X may have any value from $Ø to $F(0-15). Please take note that a dollar sign ($) before a number means the number is in HEX. The value of X determines the action the clock will take. Following is a table which shows the command for the clock.

X	COMMAND
Ø	READ $2^{2Ø}$-2^{24}TIME BITS
1	READ 2^{12}-2^{19}TIME BITS
2	READ 2^{4} -2^{11}TIME BITS
3	READ 1ØØms-2^{3} TIME BITS
4	READ 1msecs-1Ømsecs
5	START CLOCK
6	STOP CLOCK
7	ADVANCE ONE OR CLEAR IRQ
8	ADVANCE TWO OR CLEAR INT-OUT
9	SET INTERRUPT

Table 9

CLOCK COMMANDS

Suppose your Apple Clock is in Slot #4. The following table lists the addresses and commands. All commands except SET INTERRUPT should be done with a PEEK when in BASIC, or a LOAD instruction from Assembly language.

ADDRESS		
HEX	DECIMAL	COMMAND
CØCØ	-16192	READS $2^{2Ø}$-2^{24} TIME BITS
CØC1	-16291	READS 2^{12}-2^{19} TIME BITS
CØC2	-1619Ø	READS 2^{4} -2^{11} TIME BITS
CØC3	-16189	READS 1ØØms-2^{3} TIME BITS
CØC4	-16188	READS 1ms-1Ømsec TIME BITS
CØC5	-16187	START CLOCK
CØC6	-16186	STOP CLOCK
CØC7	-16185	ADVANCE ONE OR CLEAR IRQ
CØC8	-16184	ADVANCE TWO OR CLEAR INT-OUT
CØC9	-16183	SET INTERRUPT

Table 10

SLOT #4 EXAMPLE

To start the clock simply do a

PEEK(-16187) in BASIC

or LDA $CØC5 in Assembly Language

In order to stop the clock:

PEEK(-16186) in BASIC

or LDA $CØC6 in Assembly Language

READING DIGITS OF TIME

There are five digits of time ranging from 1's of milliseconds to 2^{24} seconds. The digits less than a second are in BCD format. Digits of seconds and above are in binary.

To read a digit of time either PEEK a location when in BASIC or do a LOAD instruction from that address in assembly language.

When reading the top byte of time, only the bottom 5 bits of this byte are used for time. This table shows the bit configuration for the top byte (the lowest device select address).

X = Ø

	BIT	MEANING
MSB	7	(Ø) Clock Board Interrupting (1) Not Interrupting
	6	(1) Leap Year (Ø) Not Leap Year
	5	Not Used
	4	2^{24} Seconds
	3	2^{23} Seconds
	2	2^{22} Seconds
	1	2^{21} Seconds
LSB	Ø	$2^{2Ø}$ Seconds

Table 11
TOP BYTE OF TIME CONFIGURATION

ROM FIRMWARE

The Apple Clock has an on-board ROM that allows easy access to the time. It may be accessed from BASIC or assembly language, and provides date and time information.

When in BASIC, the input routine switches may be set using the IN#n command. Following this by an INPUT statement, the ROM will be activated and send back the time to BASIC.

To stop the printout of the time from an INPUT statement, the printout switch may be set to the clock board by using the PR#n command. After reading the time, both switches should be set back to Ø for normal Apple operation (PR#Ø:IN#Ø).

When using assembly language, a call to CNØØ will give the time and place it in the following locations in the Apple input buffer:

ADDRESS	USE
Ø28Ø	Not Used
Ø281	Carriage Return
Ø282	1's milliseconds
Ø283	1Ø's milliseconds
Ø284	1ØØ's milliseconds
Ø285	.
Ø286	1's seconds
Ø287	1Ø's seconds
Ø288	;
Ø289	1's minutes
Ø28A	1Ø's minutes
Ø28B	;
Ø28C	1's hours
Ø28D	1Ø's hours
Ø28E	Space
Ø28F	1's days
Ø29Ø	1Ø's days
Ø291	/
Ø292	1's months
Ø293	1Ø's months
Ø294	Space
Ø2AØ	Counter
02A1	Temporary Storage

Table 12

ROM MEMORY USE

CHANGING ROMS

Your Apple Clock is shipped with a ROM which is a 2708 equivalent. At some later date you may wish to insert a 2716 PROM. This can easily be done by adding a few jumpers. On the PC board above the ROM notice there are numbers: 13, 11, 12, 8, 9 and 10. Follow these instructions for a 2716 PROM.





SETTING THE FREQUENCY

Your Apple Clock has been factory assembled, burned in, and tested. The 1.0000MHz time base has been accurately set to within .001%. Vibrations or extreme temperatures can cause slight changes to the time base and may produce noticeable errors. If these errors are noticed, or if you desire to set this frequency more precisely for your environment, an accurate frequency counter and a small non-metallic screwdriver are required.

Connect the frequency counter with the ground lead to Pin 7 of U9, the positive lead to Pin 6 of U9. Adjust C1 for a frequency as close to 1.0000MHz as possible. Be sure the clock is at the same operating temperature as its normal environment.


8
7
6
5
D
C
B
A
STOP
+5VC
4069
U9
C1
3.5-20 PF
R1 1K
R2, 1M
Y1
C2 18PF
C3 18PF
U1 4518
U2 4518
U3 4518
1 MSEC
10MS
100 MS
1 SEC
4503 37 PLS
+5VT
R19 1K
RESET
P1
R3 2.2K
A3
A2
A1
A0
R4 3.3K
R5 3.3K
R6 3.3K
U13 74C42
LS367
START CLOCK
STOP CLOCK
R7 10K
U10 4013
WRITE PROTECT
S1
RUN
ADV 1
ADV 2
CLR IRQ
4503
R9 10K
DMA IN
DMA OUT
SET INT
LS04
U11 4013
DEV SEL
I/O SEL
R/W
U6 LS10
U24 LS00
R8 10K
CLR INT-OUT
INT IN
A7
A8
A9
A10
U23 LS13
I/O STB
GND
+5VDC
CLR-PROM
PROM-ON
C9 15UF
C10 .01UF
C11 .01UF
C12 .01UF
NOTES: UNL
1. ALL RESI
OHMS, 1/
2. U14 - U21
U22, U25,
PIN 8 = GN
REF DES
LAST USED
NOT USED
BT1
C12
CR2
P1
Q2
R19
S1
U28
Y1
13

-S OTHERWISE SPECIFIED-

OR VALUES ARE IN
, 5%.
503) - PIN 16 = + 5VC, PIN 8 = GND.
26 (74LS367) - PIN 16 = +5VT,

UNLESS OTHERWISE SPECIFIED	CONTRACT NO.
XX DEC ± 010	
XXX DEC ± .005	DO NOT SCALE DRAWING
FRACTIONS ± 1/64	DR LORETTA WILLE 09 28 78
ANGLES ± 0° 30'	CHK
MATERIAL	PROJ ENGR
TREATMENT	
FINISH RMS	MFG ENGR

14-000070
NEXT ASSY | USED ON
APPLICATION

Mountain Hardware, Inc.

SCHEMATIC DIAGRAM - APPLE CLOCK

SIZE	CODE IDENT NO.	DRAWING NO.
D		20-000071

SCALE | WT | SHEET 1 OF 1

DRAWING NO.

WARRANTY

Your factory-built Apple Clock is warranted against defects in materials and workmanship for a period of six (6) months from the date of delivery. We will repair or replace products that prove to be defective during the warranty period, provided they are returned to Mountain Hardware, Inc. No other warranty is expressed or implied. We are not liable for consequential damages. We reserve the right to refuse to repair any product that in our opinion has been subjected to abnormal electrical or mechanical abuse. Products out-of-warranty are subject to a minimal service fee.

Please feel free to contact us if you have any questions or problems.